NAG Library Routine Document

C05ADF

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

C05ADF locates a zero of a continuous function in a given interval by a combination of the methods of nonlinear interpolation, extrapolation and bisection.

2 Specification

```
SUBROUTINE C05ADF(A, B, EPS, ETA, F, X, IFAIL)
INTEGER            IFAIL
double precision   A, B, EPS, ETA, F, X
EXTERNAL           F
```

3 Description

C05ADF attempts to obtain an approximation to a simple zero of the function $f(x)$ given an initial interval $[a, b]$ such that $f(a) \times f(b) \leq 0$. The same core algorithm is used by C05AZF whose specification should be consulted for details of the method used.

The approximation x to the zero α is determined so that at least one of the following criteria is satisfied:

(i) $|x - \alpha| \leq 4 \times \text{EPS} \times \max(|x|, 1)$,

(ii) $|f(x)| \leq \text{ETA}$.

4 References

Brent R P (1973) *Algorithms for Minimization Without Derivatives* Prentice–Hall

5 Parameters

1: A – ***double precision*** *Input*

On entry: a, the lower bound of the interval.

2: B – ***double precision*** *Input*

On entry: b, the upper bound of the interval.

Constraint: $\text{B} \neq \text{A}$.

3: EPS – ***double precision*** *Input*

On entry: the termination tolerance on x (see Section 3).

Constraint: $\text{EPS} > 0.0$.

4: ETA – ***double precision*** *Input*

On entry: a value such that if $|f(x)| \leq \text{ETA}$, x is accepted as the zero. ETA may be specified as 0.0 (see Section 7).

5: F – ***double precision*** FUNCTION, supplied by the user. *External Procedure*

F must evaluate the function f whose zero is to be determined.

The specification of F is:

```
double precision FUNCTION F(XX)
double precision            XX
```

1: XX – ***double precision*** *Input*

On entry: the point at which the function must be evaluated.

F must be declared as EXTERNAL in the (sub)program from which C05ADF is called. Parameters denoted as *Input* must **not** be changed by this procedure.

6: X – ***double precision*** *Output*

On exit: the approximation to the zero.

7: IFAIL – INTEGER *Input/Output*

On entry: IFAIL must be set to 0, -1 or 1. If you are unfamiliar with this parameter you should refer to Section 3.3 in the Essential Introduction for details.

On exit: IFAIL $= 0$ unless the routine detects an error (see Section 6).

For environments where it might be inappropriate to halt program execution when an error is detected, the value -1 or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, if you are not familiar with this parameter, the recommended value is 0. **When the value -1 or 1 is used it is essential to test the value of IFAIL on exit.**

6 Error Indicators and Warnings

If on entry IFAIL $= 0$ or -1, explanatory error messages are output on the current error message unit (as defined by X04AAF).

Errors or warnings detected by the routine:

IFAIL $= 1$

On entry, EPS ≤ 0.0,
or A $=$ B,
or F(A) $\times$ F(B) > 0.0.

IFAIL $= 2$

Too much accuracy has been requested in the computation; that is, the interval containing the zero has been reduced to one of relative length at most 2ϵ, where ϵ is the ***machine precision***, but the exit conditions described in Section 3 are not satisfied. It is unsafe to continue reducing the interval beyond this point, but the final value of X returned is an accurate approximation to the zero.

IFAIL $= 3$

A change in sign of $f(x)$ has been determined as occurring near the point defined by the final value of X. However, there is some evidence that this sign-change corresponds to a pole of $f(x)$.

7 Accuracy

The levels of accuracy depend on the values of EPS and ETA. If full machine accuracy is required, they may be set very small, resulting in an exit with IFAIL $= 2$, although this may involve many more iterations than a lesser accuracy. You are recommended to set ETA $= 0.0$ and to use EPS to control the accuracy, unless you have considerable knowledge of the size of $f(x)$ for values of x near the zero.

8 Further Comments

The time taken by C05ADF depends primarily on the time spent evaluating F (see Section 5).

If it is important to determine an interval of relative length less than $2 \times$ EPS containing the zero, or if F is expensive to evaluate and the number of calls to F is to be restricted, then use of C05AZF is recommended. Use of C05AZF is also recommended when the structure of the problem to be solved does not permit a simple F to be written: the reverse communication facilities of C05AZF are more flexible than the direct communication of F required by C05ADF.

9 Example

This example calculates an approximation to the zero of $e^{-x} - x$ within the interval $[0, 1]$ using a tolerance of EPS = 1.0D−5.

9.1 Program Text

```
*     C05ADF Example Program Text
*     Mark 14 Revised. NAG Copyright 1989.
*     .. Parameters ..
      INTEGER          NOUT
      PARAMETER        (NOUT=6)
*     .. Local Scalars ..
      DOUBLE PRECISION A, B, EPS, ETA, X
      INTEGER          IFAIL
*     .. External Functions ..
      DOUBLE PRECISION F
      EXTERNAL         F
*     .. External Subroutines ..
      EXTERNAL         C05ADF
*     .. Executable Statements ..
      WRITE (NOUT,*) 'C05ADF Example Program Results'
      A = 0.0D0
      B = 1.0D0
      EPS = 1.0D-5
      ETA = 0.0D0
      IFAIL = 1
*
      CALL C05ADF(A,B,EPS,ETA,F,X,IFAIL)
*
      WRITE (NOUT,*)
      IF (IFAIL.EQ.0) THEN
         WRITE (NOUT,99999) 'Zero at X =', X
      ELSE
         WRITE (NOUT,99998) ' ** C05ADF returned with IFAIL = ', IFAIL
         IF (IFAIL.EQ.2 .OR. IFAIL.EQ.3) WRITE (NOUT,99999)
     +       'Final point = ', X
      END IF
*
99999 FORMAT (1X,A,F12.5)
99998 FORMAT (1X,A,I5)
      END
*
      DOUBLE PRECISION FUNCTION F(X)
*     .. Scalar Arguments ..
      DOUBLE PRECISION X
*     .. Intrinsic Functions ..
      INTRINSIC        EXP
*     .. Executable Statements ..
      F = EXP(-X) - X
      RETURN
      END
```

9.2 Program Data

None.

9.3 Program Results

```
 C05ADF Example Program Results

 Zero at X =     0.56714
```